


**MSX<sub>2</sub> MSX<sub>2+</sub>**

HIS-60

# Handy Scanner-MSX<sub>2</sub>

ハンディスキャナー-MSX<sub>2</sub>

取扱い説明書



**HAL**  
Laboratory

---

## はじめに

このたびは、弊社スキャナシリーズをお買い求めいただきまして、誠にありがとうございます。  
でございます。

本書は、スキャナとスキャナカートリッジを十分に活用していただけるように、操作方法の全般を説明しています。お使いになる前に必ずお読みください。

### お客様へのお願い

- 1 システムのバックアップ以外の目的で、本ソフトウェアの一部または全部を複製・流用することを禁じます。
- 2 本マニュアルの一部または全部を複製・流用することを禁じます。
- 3 本マニュアルの内容については、予告なく変更することがあります。
- 4 ソフトウェアおよびマニュアルの内容については万全を期しておりますが、万一お気づきの点やご不明の点がございましたら、お手数ですが、弊社窓口までご一報くださるようお願いいたします。
- 5 運用した結果については、上記4項にかかわらず、責任を負いかねますのでご注意ください。
- 6 本マニュアルに掲載した画面表示・説明図などは、実際のものとは一部異なることがあります。ご了承ください。

---

### 本書の読み方

本書は、スキャナを使って画像を取り込んで編集し、利用するまでの操作全般を説明しています。

お手持ちのアプリケーションソフトについては、それぞれの取り扱い説明書をお読みください。

### 表記上の約束

- ◆キーボードのキーは枠で囲んで表記しています。
  - 「なにかキーを押してください」と説明されているときは、キーボードのキー（どれでもよい）を押してください。
  - 「Lキーを押してください」と説明されているときは、画面で指定されたキー（この場合ではL）を押してください。
  
- ◆「選択する」と書かれているときは、目的の項目や値を反転させ、リターンキーを押してください。

---

## 目次


セット内容を確認する	6
本体各部の名称と働き	7
接続のしかた	8
動作チェック	10
スキャナご使用上の注意	11
付属ソフトの解説	13
HALNOTE用スキャナプログラムの使い方	14
●付属のプログラムをHALNOTEのディスクに複写する	15
●立ち上げ方	16
●実際に取り込んでみましょう	17
●コマンド説明	18
●HALNOTE「作画」アイコン説明	19

---

BASICサンプルプログラムを使う場合	20
●実際に取り込んでみましょう	21
●各コマンド説明	22
カラー・グラフィック・エディターVer.2.00で使う場合	26
●画面のセーブとロード	27
●圧縮された画面のセーブとロード	30
●BSAVEされていないファイルに対しての傾向と対策	33
●カラー・グラフィック・エディターコマンド一覧	34
●カラー印字の方法とカラーハードコピープログラム	36
エラーメッセージ一覧	38
付録：プログラムリスト	41

## セット内容を確認する

お求めになりましたらすぐにセットの内容をご確認ください。万一、不足しているものがありましたら、弊社までご連絡ください。



### ご注意

カートリッジは、静電気などで壊れることがありますので、端子分には絶対に手を触れないでください。

本品に同梱されているディスクには大事なプログラムが入っています。大切に取り扱い、磁石等に近付けないようにしてください。

### ■セット内容の他に必要なもの

本品を画像データの取り込みだけにお使いの方は、まったく必要ありませんが、「HALNOTE」及び「グラフィックエディター」のデータとして扱う場合には、MSX-DOSまたは、MSX-BASICでフォーマットされたディスクが数枚必要です。


### ■マウスご利用のおすすめ

添付されているプログラムは、まったくマウスを必要としませんが、本品ならびに本品で作成されたデータを「HALNOTE」や「グラフィックエディター」のデータとして扱う場合は、マウスがあれば、より簡単な操作で扱うことができますようになります。各ソフトの豊富な機能をすべて利用するためにも、なるべくマウスでのご利用ください。

マウスは各社製のもののご使用になれますが、弊社からも「コボウズ (HTB-60)」が発売されてます。

マウスをコンピュータ本体に接続する方法や取り扱いについては、マウスについている取り扱い説明書をご覧ください。

## 本体各部の名称と働き



### ①DPI切り替え

図版の読み取る精度の調整をします。2段階あり、8ドットは粗く、16ドットは細かく読み取られます。8ドットを指定すると図版を読み取る範囲が大きく取れ、16ドットを指定すると読み取る範囲が半分になります。

### ②BRIGHT (ブライツ)

原稿が明るい (白の比率が多い) ときは右に、暗い (黒の比率が多い) ときには左に回して調整します。

### ③DITHER (ディザ)

B/W…イラストや文字など、白黒 (濃淡) がはっきりしている図版を読み取る場合に設定します。

DITHER…写真など白黒 (濃淡) がはっきりしない図版を読み取る場合に設定します。

ディザには、a、b、cのスイッチ設定があり、それぞれ異なった表現力で読み取ることができます。

a: 明るい部分から暗い部分までの階調がなめらかで、最も忠実に図版を読み取ります。

b: 図版のトーンが強調され、力強さがあります。

c: 粒子が粗くなり、離れたところから見るときに有効です。

\*SCREEN 6,7及びHALNOTEではb,cのディザは使用できません。

### ④CONTRAST (コントラスト)

読み取る図版のコントラスト (明暗の度合い) を調整します。

左側にするとコントラストが強くなり、右側にするとコントラストが弱くなります。中間には■の目印があります。

### ⑤読み取り位置確認窓

読み取り位置を確認するための窓です。

### ⑥スタートスイッチ

絵や写真を読み取るときに1回押します。これを押してから本体を動かせば、原稿を読み取ることができます。

### ⑦ゴムローラ


スキヤナが移動したことを検出する部分です。

### ⑧読み取り窓


図版を読み取る部分です。

## 接続のしかた

スキャナカートリッジは、スキャナで取り込んだ絵のデータをコンピューター本体に渡す重要な部分です。次の手順にしたがって正しく取り付け、スキャナ本体、ACアダプタとの接続をしてください。





- 
- 
- 1 MSX本体の電源をオフにして下さい。
  - 2 スロットに、スキャナカートリッジを差し込んでください。  
HALNOTEカートリッジと同時に使用するときは、HALNOTEカートリッジを必ずスロット1に差し込みます。
  - 3 下の図のように、ACアダプタのDCプラグを電源ジャックに、スキャナの接続コネクタをスキャナ接続端子に差し込んでください。  
  
(スキャナの接続コネクタは、矢印のついている側を手前にして差し込んでください。)
  - 4 ACアダプタをコンセントに差し込んでください。  
  
(長時間使用しないときや、DCプラグをスキャナカートリッジから抜き差しするときは、コンセントから抜いておいてください。)

**ご注意**

☆コンピュータの電源が入っているときに、カートリッジやコネクタ類の抜き差しは絶対にしないでください。

## 動作チェック

スキャナカートリッジには、BASICから簡単にスキャナを使うための拡張BASICステートメント処理ルーチンが内蔵されています。

### [形式]

```
CALL SCAN ( ST, XP, YP, XS, YS, SW, C1, C2 )
```

### [パラメータの説明]

- ST スキャナタイプを示す数値です。  
0 : HIS-10  
2 : HIS-10Y  
4 : HVS-10
- XP, YP 画面上の取り込みエリア原点 (左上) を指定します。  
※上記の範囲が小さすぎるとCan't CONTINUEが出ることがあります。
- XS, YS 画面上の取り込みエリアの大きさを、ドット数で指定します。
- C1 取り込み画像の黒ドット (図形) に対応するVRAMカラーコード  
C2 取り込み画像の白ドット (背景) に対応するVRAMカラーコード
- SW 取り込み面積を指定する数値  
1 : 拡大  
2 : 通常  
※SCREEN 5,8の場合は、2にしてください。

### ●エラーコード

コード	エラーメッセージ	エラー内容	原因
(21)	NO RESUME	ACアダプターが異常です	外部電源が切れています
(17)	Can't CONTINUE	途中で取り込みをやめた	<input type="text" value="ESC"/> キーで中断した

## スキャナご使用上の注意

スキャナのご使用にあたっては、次のような点にご注意ください。

- 直射日光のあたる場所、振動の激しい場所、湿気やほこりの多い場所、非常に高温な場所ではご使用にならないでください
- スキャナ本体の分解、改造は絶対に行わないでください。
- スキャナ本体の読み取り窓に異物を入れないでください。
- スキャナ本体の外側を、ベンジン、シンナー、化学ぞうきんで拭かないでください。汚れのひどい場合は、中性洗剤を水で薄めたものにきれいな布を浸し、きつく絞ってご使用ください。
- 落としたりぶつけたりして、スキャナ本体に強いショックを与えないでください。
- ゴムローラに汚れがついた場合は、粘着テープ等で汚れを取ってください。
- ケーブル類は、引っ張ったり無理に曲げたりしないでください。
- ケーブル類の接続・取り外しは、必ずコンピューター本体の電源が切れていることを確認し、コネクタ部分を持って行ってください。
- カートリッジの端子部には絶対手を触れないでください。

## 付属ソフトの解説

本品付属のディスクには、ハンディスキヤナの機能を理解して頂くための簡単なプログラムが納められています。また、使用にあたっては、バックアップをとったディスクでご使用することをお勧めします。

### SCAN. BAS

各部動作チェックを行い、SCANNER. BASをスタートさせるプログラムです。

### SCANNER. BAS

各スクリーンモードに対応した画像取り込みのサンプルプログラムです。スキヤナの機種指定、スクリーンモード毎の取り込み、簡単な画像編集、BASIC形式でのSAVE/LOADなどが実行可能です。

また、ユーザーの応用ソフトの制作の参考となるように、プログラムの各所にコメントが記されています。あわせて、本書42ページより掲載されているプログラムリストもご活用下さい。

### SCAN. (SC

HALNOTE用の画像取り込みプログラムです。HALNOTEのシステムディスクにプログラムを複製してお使い下さい。また、詳しい使用方法は、本書16ページ「HALNOTE用スキヤナプログラムの使い方」を参考にして下さい。

### PACKBIT. HN

HALNOTE上でスキヤナを使う時に必要なシステムプログラムです。HALNOTEのディスクにプログラムを複製してお使いください。

## HALNOTE用 スキャナプログラムの使い方

本商品は、HALNOTEとの併用を考慮して作られています。付属のプログラムをHALNOTEのディスクに複写することにより、いままでできなかった画像の取り込みや編集、文書データへの画像組み込み等が可能になります。

「ワープロ」や「図形プロセッサ」でご利用になる場合は、チューニングバー  
ジョンの道具が必要になります。

HALNOTEユーザ登録後、弊社ユーザーサポート係までお問い合わせく  
ださい。

HALNOTEを起動する前に、以下のことをチェックしてください。

■HALNOTEカートリッジがスロット1に、スキャナカートリッジがス  
ロット2に差し込まれていますか？

■スキャナカートリッジには、DCプラグが、ACアダプターはコンセント  
に差し込まれていますか？

■ハンディスキャナは、カートリッジのスキャナ接続端子に接続されていま  
すか？

■付属のプログラムをHALNOTEのディスク複写しましたか？

もし正常に動作しなかった場合は、直ちにMSX本体の電源を切り、以上の事をも  
う一度チェックして下さい。

---

## 付属のプログラムを HALNOTEのディスクに複製する

---

本商品は、付属のプログラムをHALNOTEのディスクに組み込むことによって、HALNOTE上での画像取り込みが可能になります。  
ここでは、そのプログラムを組み込む方法を、わかりやすく説明します。

●では、実際に組み込んでみましょう！

本体に付属しているディスクを入れて、キーボードから以下のように打って下さい。

```
COPY "A:SCAN.(SC" TO "B:SCAN.(SC"
```

1ドライブでお使いの場合は、もとのデータが読み込まれた後、以下のメッセージが表示されます。

```
Insert diskette for drive b:  
and strike a key when ready
```

ここで、HALNOTEのディスクを入れてリターンキーを押します。

あとは、画面に表示されるメッセージに従い本体付属のプログラムディスクとHALNOTEのディスクを数回入れ替えれば、作業は完了です。また、同じようにPACKBIT.HNも複製しておいてください。

### MSX-DOSをお持ちの方へ

付属のプログラムは、MSX-DOSを使っても複製することができます。詳しいやり方は、MSX-DOSのリファレンスマニュアルか、関連書籍をご参照ください。

---

---

## 立ち上げ方

---

1. バインダからメニューバーの” 道具” を左クリックして、道具の一覧を表示させます。
2. ” SCAN” を左クリックすれば、プログラムが立ち上がります。

### ご注意

HALNOTEは、スロット1にHALNOTEカートリッジ、スロット2にスキャナカートリッジが差し込まれてないと、正常動作しません。プログラムを実行する前に、もう一度確認してください。

---

## 実際に取り込んでみましょう

---

では、実際に取り込んでから保存するまでを、順を追って説明します。

1. メニューバーから”取込”を左クリックして、スキャナの機種を選択と、取り込み方法を決めます。
2. ”開始”を左クリックして、スキャナを原稿の上でなめらかに滑らせます。  
(1画面を数秒で取り込む程度が最適です)
3. 満足のいく画面ならば、タイトルメニューを左クリックして、”保存”を左クリックします。
4. 好きなファイルネームをつけて、リターンキーを押します。

以上が、スキャナを使って画像を取り込んでから保存するまでの、一連の操作です。



---

## HALNOTE「作画」アイコン一覧表

---



・・・「編集」などの処理対象となる範囲を指定します。基本的には「製図道具」の範囲指定とほぼ同じです。



・・・文字の入力セルを開きます。入力位置や字体等が良ければマウスを右クリックすると範囲が消えて、文字が残ります。



・・・直線を引くときに使用します。



・・・四角形を描くときに使用します。



・・・角の丸い四角形を描くときに使用します。



・・・丸を描くときに使用します。



・・・始点と終点までの多角形を描くときに使用します。  
(終点で2回クリックすると始点と終点が結ばれます)



・・・マウスで動かしたように自由な線を引くときに使用します。



・・・描かれているものを消しゴムで消すように使用します。



・・・点を打つときに使用します。  
(押したまま動かすと自由線も描けます)

### その他の機能

ノイズ消去・・・総選択やセレクトで選択された部分のスクアナ取り込み画像のノイズを消去する機能です。(ノイズ消去する前に一旦「保存」しておく方が良いと思います)

## BASICサンプルプログラムを使う場合

ここでは、本品に同梱されているディスクに納められているプログラムをもとに、実際にハンディスクャナを使って画像を取り込む方法を説明していきます。

### サンプルプログラムを起動する前に……

プログラムを起動する前に、以下のことをチェックして下さい。

- スキャナカートリッジは、スロットに差し込まれていますか？
- スキャナカートリッジには、DCプラグが、ACアダプターはコンセントに差し込まれていますか？
- ハンディスクャナは、カートリッジに接続されていますか？

もし正常に動作しなかった場合は、直ちにMSX本体の電源を切り、以上の事をもう一度チェックして下さい。

### 画像を取り込む前に……

#### ▲起動方法

BASICを起動してから、サンプルプログラムの入ったディスクをドライブAに入れて下さい。

キーボードから、RUN "SCAN, BAS" と入力し、リターンキーでスタートさせてください。以下のメニュー画面（以下ヘルプ画面）が現れます。

もしエラーが発生した場合は、カートリッジの接続不良が考えられます。MSX本体の電源を切って、もう一度接続を確認して下さい。（メニュー画面参照）

#### ▲起動時の操作

最初に表示されるヘルプ画面では、各キーに対応したコマンドの簡単な説明が記されています。また、このヘルプ画面は、何かキーを押す事で終了する（消す）ことができます。

ヘルプ画面が終了したら、スキャナのタイプを設定します。詳しくは「<T>スキャナタイプの設定」で説明します。

これで、画面に「コマンドを入力して下さい」と表示されたら準備は完了です。

---

## 実際に取り込んでみましょう

---

さあ、用意はできましたか？ では、実際に順を追って画像を取り込んでみましょう。

1. ヘルプ画面から何かのキーを押して、コマンド待ちの状態して下さい。ここでスペースキーを押すと、画面の枠が黒くなり画像取り込みの待機状態になります。（スキヤナの読み取り位置確認窓が点灯します）
2. 取り込みたい原稿の上にスキヤナをセットして、スタートスイッチを押して下さい。スキヤナを静かに滑らせれば、原稿が読み取られていきます。（1画面を数秒で取り込む程度が最適です）
3. 満足のいく取り込みができましたか？ 現在表示されている画面でよければ、「よろしいですか？」の問いに” Y” と入力して下さい。もし、満足がいかなければ” N” と入力して、1からやり直して下さい。
4. いま取り込んだ画面をディスクにセーブしましょう。” S” とキー入力したら画面の指示にしたがって、ファイルネームを入力して下さい（最高8文字まで）。ファイルネームを入力し終わったら、リターンキーを押して下さい。これで、あなたが取り込んだ画像がディスクに記録できたはずですよ。
5. では、実際にセーブできたのかどうか、ロードして確認してしてみましょう。  
” L” キーを押すと、現在ディスクに納められている画像ファイルの一覧が表示されます。目的のファイルが見つかりましたか？ では、” > LOAD ファイル=” の問いにさきほど記録したファイルネームを打ち込み、リターンキーを押してみてください。ディスクを読みにいき、さきほど記録した画像が表示されましたね。

これが実際に画像を取り込むまでの一連の操作です。ここで紹介したのは一番簡単な使用方法です。このやり方を利用して、いろんな画像を取り込んでみてください。

---

## 各コマンド説明

---

さきほどは簡単な取り込みの方法を説明しましたが、ここではより高度な取り込み・編集をするためのコマンド説明をします。これから説明するコマンドを覚えていただければ、よりいっそう簡単に、かつ複雑な編集処理が可能になります。

### **SPACE** 画像取り込み開始

スペースキーを押すことで、スキャナの画像読み取り位置確認窓が点灯し、画像取り込み待ちの状態になります。スキャナ側のスタートボタンを押すと読み取り位置確認窓がさらに明るく点灯します。あとは、原稿の上をゆっくりと滑らせればOKです。

もし、スキャナの画像読み取り位置確認窓が点灯しない場合は、以下のことが考えられます。

- (1) カートリッジが正しく接続されていない。
- (2) スキャナのプラグがカートリッジに正しく接続されていない。
- (3) カートリッジに電源が供給されていない。

取り込みが終了すると「よろしいですか?」と表示されます。満足のゆく画像であれば” Y”を入力して確定します。失敗ならば” N”を入力すれば、以前の画面が復活します。

### **ESC** 画像取り込み中止

スペースキーを押して画像取り込み待ちをしている時にこのキーを押せば、取り込みを中止して以前の画面を復活させることができます。

### **S** **F1** 画像データのセーブ

ディスクに保存されているデータの一覧が表示されて、ファイルネームの入力モードになります。

ファイルネームを入れてリターンキーを押すと、画像データをディスクに保存します。また、既に存在するファイルネームを入れた場合は、確認を求めてきます。

---

---

OKならば” Y ” と入力して下さい。ただし、この場合はすでにあるファイルにオーバーライト（重ね書き）しますので、以前のデータは失われますのでご注意ください。拡張子はプログラムが各スクリーンモードを準じて付けますので、指定しないで下さい。

### **L** **F 2** 画像データのロード

ディスクに保存されているデータの一覧が表示されて、ファイルネームの入力モードになります。

ファイルネームを入れてリターンキーを押すと、指定されたデータを読み込んで、画面に表示します。

ファイルの形式は、BASICのBSAVE/BLOADと同じ形式です。

### **V** **F 5** メッセージの消去

” V ( F 5 ) ” を押すと、「コマンドを入力して下さい」のメッセージが消えて、全画面が見えるようになります。また、この状態では全てのコマンドを受け付けません。何かキーを押せば、コマンド待ちの状態に戻ります。

### **T** **F 6** スキャナタイプの設定

” T ” を押してスキャナタイプの変更を行います。

カーソルで反転表示を左右に移動し、リターンキーで決定します。セットアップ時は” HIS-10 ” に設定されていますが、他のスキャナを使用するときは、必ずこのコマンドでスキャナのタイプを変更してからお使いになるようにして下さい。タイプの変更をしなければ、正常な動作は保証されません。

## **M** **F 7** スクリーン・モードの設定

” M ” を押してスクリーン・モードの変更を行います。  
カーソルで反転表示を左右に移動し、リターンキーで決定します。また、スクリーン・モードが変更されると、現在表示されてる画像データはクリアされるので注意して下さい。

### [各スクリーン・モードの特徴]

SCREEN 5 256×212ドット スキャナ取り込み部の左半分が有効  
SCREEN 6 512×212ドット スキャナ取り込み部の全体が有効  
SCREEN 7 512×212ドット スキャナ取り込む部の全体が有効  
SCREEN 8 256×212ドット スキャナ取り込み部の左半分が有効

## **G** **F 8** 表示ロジック（方法）の変更

” G ” を押して表示ロジックの変更を行います。  
カーソルで反転表示を左右に移動し、リターンキーで決定します。  
ロジックの意味は、BASICのCOPYステートメントと同じです。また、演算の対称になるのは、黒で表示されるドットのみです。

### [各モードの説明]

PSET 論理演算はしません。  
AND カラーコードの論理積  
OR カラーコードの論理和  
XOR カラーコードの排他的論理和

## **R** **F 9** 画面の縮小

” R ” を押して縮小の大きさの変更を行います。  
カーソルで反転表示を左右に移動し、リターンキーで実行します。縮小終了後、「よろしいですか？」の問いに” Y ” と入力すれば、完了です。もし、気に入らな

---

---

ければ” N” と入力して下さい。実行以前の画像が表示されます。

[各モードと実際に表示される大きさ]

1 / 2 縦, 横共に2分の1の大きさにします。

1 / 4 縦, 横共に4分の1の大きさにします。

### **?** ヘルプ画面の表示

コマンド待ちの状態では” ?” を押すと、ヘルプ画面を表示します。ただし、コマンドを実行して何らかの形でユーザーからの入力を要求している時は、実行されません。

### **Q** プログラムを終了する

” Q” を押すと画面に「終了しますか?」と表示されます。終了してもよければ” Y” を、編集を継続するなら” N” を押して下さい。

※説明中の各メッセージは漢字で書かれていますが画面上には、MSXのANKで表示されます。

## カラー・グラフィック・エディター Ver2.00で使う場合

ここでは、付属のプログラムで取り込んだ画像データを、「カラー・グラフィック・エディター Ver. 2.0」のデータとして扱う方法を説明します。

### カラー・グラフィック・エディターを 立ち上げる前に……

ここから先は、1ドライブ仕様のMSXをお持ちの方を前提に説明します。2ドライブ仕様のMSXをお持ちの方は、「●2ドライブのMSXで使用するには……」からお読み下さい。

#### ●1ドライブのMSXでの使用方法

1ドライブの方は、ロード及びセーブのコマンドを実行する場合のみ、Bディスクアイコンを指定しなければなりません。これは、カラー・グラフィック・エディターがAドライブをシステム読み込み用として確保してしまうため、ディスクの入れ替えができないからです。よって、Bドライブアイコンをクリックすることは、ドライブAを仮想Bドライブとして使用するために必要な作業なのです。

また、以後の説明ではドライブの指定を行いませんので、お手持ちのMSXにあわせて判断してください。

#### ●2ドライブのMSXで使用するには……

2ドライブのMSXをご使用の方は、2ドライブがデータ専用のドライブとして割り当てられます。カラー・グラフィック・エディターのディスクをドライブ1に入れて、データの入ったディスクをドライブ2に入れてリセットして下さい。

#### ●カラー・グラフィック・エディターのデータとして扱うには

では、実際にカラー・グラフィック・エディターのデータとして扱ってみましょう。ドライブ1に付属のディスクを入れて、リセットして下さい。

タイトル画面が出たら、入力機器を選択して下さい。

※キーボードの場合、選択キーは **F1**。キャンセルキーは **F2** です。



---


## 画面のロードとセーブ

---

### ●実際に読み込んでみましょう

グラフィックエディターが立ち上がったら編集アイコンをクリックして、各コマンドの設定を行います。


セットアップアイコンをクリックして、さらにディスクロードアイコンを選びます。



クリアメニュー ペン セットアップ

以下のウィンドが開いたら、OBASICを左クリックした後に右クリックします。

編集画面に戻ったら、ディスクロードアイコンを左クリックします。




ウィンドが開いたら、目的のファイルを左クリックし、LOADアイコンを左クリックします。(1ドライブのMSXをご使用の方は、Bドライブアイコンを左クリックした後に、ファイルを選んで下さい)

目的の画面が表示されましたか？ このように、BASICのBSAVE形式で保存されているファイルは、一度セットアップモードにして、ロードする環境を変更しなければなりません。しかし、一度環境を設定してしまえば、MSX本体の電源を切るまでは、その環境が持続されます。

---

カラー・グラフィック・エディターでは、以下のようにBASIC上で保存されたデータを扱います。



●ロードできない場合は……

ロードできない場合は以下のことが考えられます。

Q ファイルの一覧が表示されない。

A ファイルがセーブされたスクリーンモードと、カラー・グラフィック・エディターのスクリーンモードが一致していないことが考えられます。画像データは、同じスクリーンモードでないとロードできないので、セットアップでデータと同じスクリーンモードに変更して下さい。

Q 画像が最後まで表示されない。

A ディスクの内容が破壊されているか、コピーする時にディスクを入れ替える回数を間違えたかのどちらかです。1ドライブの人は、もう一度複写しなおして下さい。2ドライブの人は、完全に内容が破壊されています。残念ですが、そのデータはあきらめて下さい。

---

## ●セーブしてみましょう


現在表示されている画像を、BASICのBSAVE形式でセーブしてみましょう。

編集アイコンを左クリックした後、セットアップアイコンを左クリックして、環境設定モードに入ります。

ディスクセーブアイコンを左クリックして、OBASICを左クリックします。  
(セーブ環境は、本体の電源を切るまで持続されます)  
右クリックして編集モードに戻し、ディスクセーブアイコンを左クリックします。

キーボードからファイルネームを打ち込んで、リターンキーを押せば、ディスクに保存されます。(すでに存在するファイルネームを打ち込むと、確認を求めてきます。よろしければ"OK"のアイコンを左クリックして下さい。ただし、この場合はオーバーライトをしますので、以前のデータは失われます)

カラー・グラフィック・エディターでは、以下のようにBASIC上で扱えるファイルとして保存します。



以上が、BASICで保存されたファイルの扱い方法です。

---

---

## 圧縮された画面のセーブとロード

---

グラフィックエディターには、画像データをコンパクトにまとめてセーブする機能があります。これを利用することにより、より多くの画像データを一枚のディスクに納めることができます。

ここでは、画像データをコンパクトにまとめてセーブ（圧縮セーブ）する方法を説明します。

### ●圧縮をかけたセーブ方法

まず、ロード等を行い、画面に画像を表示させます。（画面1枚必要）

編集モードからセットアップして、ディスクセーブアイコンを左クリックします。（画面1枚必要）

○G-E d i t o rを左クリックし、右クリックして編集モードに戻します。  
（立ち上げ時は、G-E d i t o rになっているので、変更の必要はありません）  
（画面1枚必要）


あとは、B A S I Cモードの時と同じように操作して、セーブして下さい。

### ●セーブ時の注意

圧縮セーブをする時は、スクリーンモードに関係なく、拡張子が” S C C ” に統一されます。よって、すでに別のスクリーンモードで現在セーブしようとしているファイルと、同じファイルネーム指定してセーブされているファイルが存在する場合は、オーバーライト（重ね書き）をしてしまうので、十分に注意が必要です。

---


カラーグラフィックエディターでは、以下のような手順で圧縮セーブを行います。



### ●圧縮された画像データのロードのしかた

圧縮されたファイルをロードするには、通常の方法と同じようにセットアップし、環境変更で `OG-Editor` を左クリックして下さい。あとは、通常のロードと同じです。

カラーグラフィックエディターでは、以下のように圧縮されたデータをロードします。



---

## ●圧縮セーブの利点

圧縮をかけてセーブすると、ファイルサイズはBASICのBSAVE形式の約1/3～1/5の大きさになります。したがって、BSAVE形式のファイルを扱う時よりも、3～5倍ちかいデータを1枚のディスクに保存しておけるわけです。

また、圧縮をかけて保存しておけば、カラー・グラフィック・エディターに読み込ませることによって、いつでもBSAVE形式のファイルにもどすことができます。

## ●カラー・グラフィック・エディターを活用する

カラー・グラフィック・エディターのデータとして扱うことで、より一層きれいなものに仕上げてみてはどうでしょうか？

- ☆ スクリーンパレットの変更をして、より原画に近いものに！
- ☆ 作画でよりアートなものにする。
- ☆ 編集で細かい手直しをする。
- ☆ 文字入力を使えば、タイトルの作成だって簡単！
- ☆ カラープリントアウトも可能！

---

## BSAVEされていないファイルに対しての 傾向と対策

---

BSAVE以外にMSX2では、COPYの命令を使って絵を保存する方法があります。その場合、たとえ拡張子（ファイル名のうしろの半角3文字）が同じでも、BLOADしているソフトウェアでは読み込むことができません。

これを読み込むようにするには、下記のようなプログラムでBSAVEしなおす必要があります。

ここでは例として数行のプログラムと、対応しているSCREENのアドレスを記載しておきますので、お手持ちのBASICリファレンスか市販されている書籍等と照らし合わせてBSAVEを行ってください。

```
10 SCREEN 8
20 COPY "FILENAME. SC8" TO (0,0)
30 BSAVE "FILENAME. SC8",0,&HD3FF,S
```

BSAVEの書式は次の書式で

BSAVE "FILENAME. SCX", [スタートアドレス], [エンドアドレス], S













各SCREENのスタートアドレスは[0]で、エンドアドレスは一般的に以下の番地です。

```
SCREEN 5      &H69FF
SCREEN 6      &H69FF
SCREEN 7      &HD3FF
SCREEN 8      &HD3FF
```















---

## カラー・グラフィック・エディター コマンド一覧

---

作画用コマンド	機能
	押している間だけ曲線を引きます
	画面上にある色を使って曲線を引きます
	指定された2点の間に線を引きます
	指定された領域を別の場所に複製します
	押している間だけ絵を消します
	スプレーをかけます
	指定された大きさの四角形を描きます
	指定された大きさの円弧を描きます
	線で囲まれた領域を塗ります
編集コマンド	機能
	指定された部分を拡大し、修正します
	指定された領域内の指定色を他の色に変更します
	指定された領域を上下反転します



	画面を消去します
	指定した領域を回転表示します
	指定された領域を左右反転します
	画像データの読み込みをします
	ディスクを初期化します
	画像データの保存をします
	プリントアウトします
<b>作画のセットアップ</b>	
<b>機能</b>	
	スプレアの密度、大きさを指定します
	ペイント方法を変更します
	使う文字の細かい設定を行います
<b>編集のセットアップ</b>	
<b>機能</b>	
	重ね合わせの指定、ファイルのフォーマットを指定します
	ファイルのフォーマットを指定します
	プリンターの設定を行います
	スクリーンモードの変更と画面の位置を調整します

---

## プリンタ設定とカラーハードプログラム

---

### ●プリンタ設定

- SMI-720
- SMI-7020
- HBP-F1
- HBP-F1C
- HBP-F1C(1/2)
- PRN-M24
- PRN-T24

\*ブラザーのM-1024系、M-1224P/Xは、PRN-M24の設定で印刷を行ってください。

## ●カラーハードコピープログラム

### ●内容

このプログラムは、カラー・グラフィック・エディタで作ったBASICファイルを使ってカラーハードコピーするプログラムです。

キーボードからLOAD"PRINT.BAS"で読み込んでからご使用ください。

### ●対応プリンター

FS-PC1のみ。それ以外のプリンターは対応していません。

### ●スクリーンモード

SCREEN 8

#### ●注意

FS-PC1のリボンカセットはA面B面で色順番が逆になります。したがって印刷中にリボンカセットを交換する場合は、同じ面の交換をお勧めいたします。カラーハードコピーは、1画面フルに印字して約30分ほどかかります。

このハードコピープログラムは、4色のリボンを使って256色を近似的に表現しておりますので、実際の画面の色とは多少異なって印字される場合があります。

PRINT. BAS

```
10 CLEAR 300,&HD200
20 BLOAD"fs_pcl.bin",R
30 SCREEN 8
40 BLOAD"girl.sc8",S
50 CMD CPR 0,0,0
```

呼び出したいSCREEN 8の絵のBLOAD  
印刷の実行及びパラメータ指定

```
CMD CPR 0, 0, 0
          |  |  |
          |  |  └─ リボンカセットの面指定
          |  └─ 横方向のオフセット指定
          └─ 大きさ
```

### ●印刷時の大きさ

0・・・普通  
1・・・1/4が2個横並び  
2・・・1/4が1個

### ●オフセット値

ドット単位 (例 25→25ドット)

### ●リボンカセット面

0.....A面 1.....B面

## エラーメッセージと対処法

誤った操作をしたり、操作中に何か異常があるとエラーメッセージが表示されます。

以下のメッセージ一覧表をご参照ください。

(尚、一覧表に記載されていないメッセージに関しては、弊社ユーザーサポート係までご連絡ください。)

### ■HALNOTE [SCAN.(SC)] の主なエラーと対策

メッセージ	内 容	対 策
プリンタが接続されていません。	プリンタが印字できる状態ではない	プリンタの状態を確認してみる
データが正しくありません	ファイルの内容が当ソフトでは扱えない	バックアップしてあったファイルがあればそれを使用してみる
スキャナがつながれていません	スキャナの接続がおかしい	スキャナの接続を確認してみる
中止しました	スキャナ取り込みを途中で中止した	ESCキーで中止したとき等に表示される
電源が入っていません	スキャナの外部電源が供給されていない	ACアダプタの接続、コンセントの具合を確認してみる

■ BASICサンプルプログラムの主なエラーと対策

メッセージ	内 容	対 策
ファイル ガ アリマセン	該当するファイルが無い	ファイル名を確認してみる
ファイル メイ エラー	入力したファイル名が不当	BASICで扱えるファイル名にしてみる
ドライブ メイ エラー	指定したドライブ名が不当	ディスクドライブの順番を確認してみる
ディスク ガ イッパイデス	フロッピーディスクの空容量が無い	違うディスクを準備する
ライトプロテクト ヲハズシテクダサイ	フロッピーディスクが書き込み禁止になっている	フロッピーディスクのライトプロテクトを塞いで書き込み可能にする
ディスク ヲ イレテクダサイ	フロッピーディスクがドライブに入っていない	ディスクドライブにフロッピーディスクを入れる

※その他コマンド入力待ちになる場合は、各種接続及び設定を確認してください。



付録  
プログラムリスト

```

1000 ' /*-----*/
1010 ' /* SCAN.BAS      for SCANNER.BAS */
1020 ' /*-----*/
1030 ' /* Version 01.000  [1989.07.02] */
1040 ' /* 1. Test SCANNER Board      */
1050 ' /* 2. RUN"SCAN.BAS"          */
1060 ' /*-----*/
1070 ' /* Created by HAL Laboratory.  */
1080 ' /*-----*/
1090 '
1100 '
1110 ' <<< Set Up >>>
1120 DEFINT A-Z
1130 COLOR = NEW : COLOR 1,15,15 : SCREEN 0 : WIDTH 40
1140 '
1150 '
1160 ' <<< Test SCANNER Board >>>
1170 ON ERROR GOTO 1180 : _SCAN( -1 ) [ ワザト エラ- ラ ハッセイ サセル
1180 IF ERR = 5 THEN RESUME 1360      [ セイショウ ナ ハ'アイ ハ、 Illegal Funcation Call
1190 IF ERR = 2 THEN RESUME 1250      [ ホ-ト' カ' ショキカ サレテ イナイ ト、 Syntax Error
1200 COLOR 12      : PRINT"# Sorry, Something is wrong ..."
1210 ON ERROR GOTO 0
1220 END
1230 '
1240 '
1250 ' <<< SCANNER Board Not Exist >>>
1260 ON ERROR GOTO 0
1270 BEEP      : COLOR 6
1280 PRINT"# SCANNER カ-トリッチ' カ' セツク サレテ イマセン."
1290 PRINT
1300 PRINT"# MSX ト カ-トリッチ' ノ テンゲン ラ キッテ カラ."
1310 PRINT"# タクシク スロット ニ ソウチャク シナオシテ クタ'サイ."
1320 PRINT
1330 END
1340 '
1350 '
1360 ' <<< Go Ahead !! >>>
1370 ON ERROR GOTO 0
1380 PRINT"# Just a moment, please ..."
1390 GOSUB 1430
1400 RUN"SCANNER.BAS"
1410 '
1420 '
1430 ' <<< Define Function Keys >>>
1440 RESTORE 1540
1450 READ K      : IF K < 1 OR K > 10 THEN RETURN
1460 A$ = ""
1470 READ X$
1475 IF X$ = ""      THEN KEY K, A$      : GOTO 1450
1480 IF LEFT$( X$, 1 ) <> "*" THEN A$ = A$ + X$ : GOTO 1470
1490 IF LEFT$( X$, 2 ) = "**" THEN A$ = A$ + X$ : GOTO 1470
1500 A$ = A$ + CHR$( VAL( "&H" + MID$( X$, 2 ) ) )
1510 GOTO 1470
1520 '
1530 ' ---No., String      [, String .....], ""
1540 DATA 1 , "width80"      , %0D      , ""
1550 DATA 2 , "auto "      , ""      , ""
1560 DATA 3 , "goto "      , ""      , ""
1570 DATA 4 , "list "      , ""      , ""
1580 DATA 5 , "run"      , %0D      , ""
1590 DATA 6 , "color 1,15,15", %0D      , ""
1600 DATA 7 , "load"      , %16      , ""
1610 DATA 8 , "cont"      , %0D      , ""
1620 DATA 9 , "list."      , %0D      , %1E, %1E, ""
1630 DATA 10 , %0C      , "run", %0D      , ""
1640 DATA -1

```



```

1000 ' /*-----*/
1010 ' /*          SCANNER.BAS          */
1020 ' /*-----*/
1030 ' /* Version 01.000   [1989.07.30] */
1040 ' /*          */
1050 ' /*          */
1060 ' /*-----*/
1070 ' /* Created by HAL Laboratory.   */
1080 ' /*-----*/
1090 '
1100 '   ショキ セットイ
1110 MAXFILES = 1 : CLEAR 512
1120 DEFINT A-Z : ON STOP GOSUB 1430
1130 STOP ON : GM = 1
1140 OPEN "GRP:" AS #GM ' [ クラフィック カメン メッセージ
1150 CM$ = "Qq VvTtMmGgRrSsLl??" ' [ コマント テーブル
1160 FK$ = "SL.VTMGR." ' [ ファンクション キー テーブル
1170 YN$ = "YyNn" ' [ Yes-No ノ ハンディ
1180 DIM WI(2048) ' [ ウィントウ ノ カソウヲ ホソシ スル ハイレツ
1190 DIM MS$(4),MX(4),MW(4) ' [ センタク ウィントウ ノ メッセージ ト サヒョウ
1200 GOSUB 4330 ' [ SCREEN 8 カラー テーブル、 ファンクション キー セットイ
1210 SM = 5 : P1 = 1 : P2 = 15 ' [ スクリーン モード、 ON/OFF ドット ノ ハレット コード
1220 GOSUB 1900 : GOSUB 4560 : GOSUB 2810 ' [ SCREEN、 ヘルプ、 スキヤナ ノ タイフ
1230 '
1240 '
1250 ' <<< Main Loop >>>
1260 ON STOP GOSUB 1550 : STOP ON
1270 ON ERROR GOTO 3550
1280 IF WI(0) = 0 THEN MS$ = "コマント を じゅうりょく して ください" : GOSUB 2250
1290 IF FK THEN K$ = MID$(FK$,FK,1) : GOTO 1320
1300 ' ^ [ FK = ファンクショ キー No.
1310 K$ = INKEY$ : IF K$ = "" GOTO 1290
1320 K = INSTR( CM$, K$ ) -1
1330 IF K < 0 GOTO 1380 ELSE GOSUB 2450
1340 ON K#2+1 GOSUB 1550,2520,2750,2810,2880,2950,3010,3400,3470,4560
1350 FK = 0 : HP = 0
1360 GOTO 1250
1370 '
1380 PLAY"V15T120L406B" : HP = HP+1
1390 IF HP > 3 THEN HP=0:FK=0:GOSUB 4560
1400 GOTO 1280
1410 '
1420 '
1430 ' <<< Termination >>>
1440 STOP STOP : ON ERROR GOTO 1510
1450 SCREEN 0 : COLOR = NEW
1460 COLOR 1,14,14 : CLS
1470 WIDTH 40 : KEY ON
1480 FOR K = 1 TO 10 : KEY(K) OFF
1490 NEXT : CLOSE
1500 GOSUB 1710
1510 ON ERROR GOTO 0
1520 END
1530 '
1540 '
1550 ' <<< Quit >>>
1560 ON STOP GOSUB 1730 : GOSUB 2680
1570 MS$ = "じゅうりょく しますか ? (y/n)"
1580 GOSUB 1640
1590 IF OK GOTO 1430 ELSE ON ERROR GOTO 1610
1600 RETURN 1600 ' [ RETURN without GOSUB エラー テ' ヌケル
1610 RESUME 1250
1620 '
1630 '
1640 ' <<< Confirm >>>
1650 GOSUB 1710 : GOSUB 2250 : R = 0
1660 R = R +1 : IF R > 50 THEN LINE(WX,WY)-STEP(WW,WH),XO,BF,XOR : R = 0

```

```

1670 K# = INKEY# : IF K# = "" GOTO 1660 ELSE K = INSTR( YN#, K# )
1680 IF K = 0 GOTO 1660 ELSE OK = K < 3 : GOSUB 2450
1690 '
1700 '
1710 '<<< Flush Key Buffer >>>
1720 IF INKEY# <> "" GOTO 1720 ELSE RETURN
1730 '
1740 '
1750 '<<< CALL SCAN >>>
1760 COLOR,,C1
1770 ER = 0 : _SCAN( ST, XP, YP, XS, YS, SW, C1, C2 )
1780 COLOR,,C2
1790 RETURN
1800 '
1810 '*** ハ'ラメ-タ ノ セツメイ ***
1820 'ST := スキャナ ノ タイフ' ( 0, 2, 4 )
1830 'XP,YP := トリコミ エリア ノ ゲンテン サ'ヒョウ
1840 'XS,YS := トリコミ エリア ノ サイズ'
1850 'SW := トリコミ メンセキ ノ シテイ ( 1, 2 )
1860 'C1,C2 := ON/OFF カラ- コ-ト' ( カ'ゾウ ノ イロ, ハ'ック ノ イロ )
1870 'SA := V-RAM ノ セントウハンチ
1880 '
1890 '
1900 '<<< Change Screen Mode >>>
1910 STOP STOP : ON SM -4 GOSUB 2040,2090,2140,2190 ' [ ハ'ラメ-タ セツテイ
1920 C3 = C2 : GOSUB 1970 ' [ スクリーン ショキカ
1930 LO = 0 : STOP ON ' [ LOGIC ハ PSET ニ ナル
1940 RETURN
1950 '
1960 '
1970 '<<< Set Screen Mode >>>
1980 SCREEN SM,,,,0 : COLOR C1,C2,C3
1990 CLS : WI(0) = 0 ' [ ウィント'ウヲ ムコウ ニ スル
2000 IF TX = 0 THEN SET PAGE 0,1 : CLS : SET PAGE 0,0 ' [ ウラ カ'メン モ クリア
2010 GOTO 4210 ' [ SCREEN モ-ト' カ' カワッタ ノテ、ハ'レットヲ ショキカ スル
2020 '
2030 '
2040 '<<< SCREEN 5 >>>
2050 XP = 0 : YP = 0 : XS = 256 : YS = 212 : SW = 2 : AX = 256 : SA = &H69FF
2060 C1 = P1 : C2 = P2 : XO = 2
2070 RETURN
2080 '
2090 '<<< SCREEN 6 >>>
2100 XP = 0 : YP = 0 : XS = 512 : YS = 212 : SW = 2 : AX = 512 : SA = &H69FF
2110 C1 = P1 AND 3 : C2 = P2 AND 3 : XO = 2
2120 RETURN
2130 '
2140 '<<< SCREEN 7 >>>
2150 XP = 0 : YP = 0 : XS = 512 : YS = 212 : SW = 2 : AX = 512 : SA = &HD3FF
2160 C1 = P1 : C2 = P2 : XO = 2
2170 RETURN
2180 '
2190 '<<< SCREEN 8 >>>
2200 XP = 0 : YP = 0 : XS = 256 : YS = 212 : SW = 2 : AX = 256 : SA = &HD3FF
2210 C1 = CT(P1) : C2 = CT(P2) : XO = 255
2220 RETURN
2230 '
2240 '
2250 '<<< Display Message >>>
2260 IF LEN(MS#) > 30 THEN MS# = LEFT$(MS#,30) ' [ 30 モシ' イナイ ニ セイゲ'ン
2270 WW = LEN(MS#) * 8 + 5 : WH = 13 ' [ ウィント'ウ ノ ハバ' ト タカサ
2280 GOSUB 2330 : PRESET( WX+4, WY+3 ) ' [ メッセ-ジ' サ'ヒョウヲ セツテイ
2290 PRINT #GM, MS#
2300 RETURN
2310 '
2320 '
2330 '<<< Open Message Window >>>

```

```

2340 GOSUB 2450
2350 WX = (AX - WW) * 2          '[ WW := ウインドウ ハハ'
2360 WY = 212 - WH             '[ WH := ウインドウ タカサ'
2370 COPY (WX,WY)-STEP(WW,WH),0 TO WI          '[ カソウヲ ハイレッ WI ニ ホソ'ン
2380 IF SM = 8 THEN C = CT(3) ELSE C = 3
2390 COLOR ,C : LINE(WX,WY)-STEP(WW,WH),C,BF   '[ ウインドウヲ クリア
2400 IF SM = 8 THEN C = CT(1) ELSE C = 1
2410 COLOR C : LINE(WX,WY)-STEP(WW,WH),C,B    '[ クロテ' ワクヲ カク
2420 RETURN
2430 '
2440 '
2450 '<<< Close Message Window >>>
2460 STOP STOP
2470 IF WI(0) <> 0 THEN COPY WI TO (WX,WY),0    '[ カソウ フックツ
2480 WI(0) = 0 : STOP ON                       '[ ウインドウヲ ムコウニ スル
2490 RETURN
2500 '
2510 '
2520 '<<< Execute SCAN >>>
2530 UD = 1 : GOSUB 2450 : GOSUB 1710 : GOSUB 1750 : IF ER GOTO 2680
2540 IF LO = 0 GOTO 2610 ELSE K = LO : IF SM = 8 AND K < 3 THEN K = K XOR 3
2550 ON K GOTO 2560,2570,2580
2560 COPY (XP,YP)-STEP(XS,YS),1 TO (XP,YP),0,AND : GOTO 2610
2570 COPY (XP,YP)-STEP(XS,YS),1 TO (XP,YP),0,OR : GOTO 2610
2580 COPY (XP,YP)-STEP(XS,YS),1 TO (XP,YP),0,XOR
2590 '
2600 '
2610 '<<< Undo ? >>>
2620 MS$ = "よろしいですか ? (y/n)" : GOSUB 1640 : IF NOT OK GOTO 2680
2630 COPY (XP,YP)-STEP(XS,YS),0 TO (XP,YP),1 : UD = 0
2640 RETURN
2650 '
2660 '
2670 '
2680 '<<< Cancel >>>
2690 STOP STOP : IF TX THEN GOSUB 1970 : UD = 1 : TX = 0
2700 GOSUB 2450 : IF UD THEN COPY (XP,YP)-STEP(XS,YS),1 TO (XP,YP),0
2710 UD = 0 : STOP ON
2720 RETURN
2730 '
2740 '
2750 '<<< Hide >>>
2760 GOSUB 2450 : GOSUB 1710
2770 IF INKEY$ = "" GOTO 2770
2780 RETURN
2790 '
2800 '
2810 '<<< Scanner Type >>>
2820 MM = 3 : MS$(0) = "SCANNER"
2830 MS$(1) = "HIS-10" : MS$(2) = "HIS-10Y" : MS$(3) = "HVS-10"
2840 M = ST/2+1 : GOSUB 3170 : ST = M-1:ST=ST*2
2850 RETURN
2860 '
2870 '
2880 '<<< Screen Mode >>>
2890 MM = 4 : MS$(0) = "SCREEN"
2900 MS$(1) = "5" : MS$(2) = "6" : MS$(3) = "7" : MS$(4) = "8"
2910 M = SM-4 : GOSUB 3170 : IF M+4 = SM THEN RETURN ELSE SM = M+4
2920 GOTO 1900
2930 '
2940 '
2950 '<<< Logical Operation >>>
2960 MM = 4 : MS$(0) = "LOGIC"
2970 MS$(1) = "PSET" : MS$(2) = "AND" : MS$(3) = "OR" : MS$(4) = "XOR"
2980 M = LO+1 : GOSUB 3170 : LO = M-1
2990 RETURN
3000 '

```

```

3010 '<<< Resize >>>
3020 MM = 2 : MS$(0) = "RESIZE" : MS$(1) = "1/2" : MS$(2) = "1/4"
3030 M = 1 : GOSUB 3170 : IF Z < 1 THEN RETURN
3040 UD = 1 : YY = YP
3050 FOR Y=YP TO YP+YS-1 STEP M*2 : [ Y ホウコウ ノ シュクショウ
3060 COPY (XP,Y)-STEP(XS,1) TO (XP,YY) : YY = YY + 1
3070 NEXT
3080 LINE(XP,YY)-STEP(XS,YP+YS-YY),C2,BF : [ ノコリヲ クリア
3090 XX = XP
3100 FOR X=XP TO XP+XS-1 STEP M*2 : [ X ホウコウ ノ シュクショウ
3110 COPY (X,YP)-STEP(1,YS) TO (XX,YP) : XX = XX + 1
3120 NEXT
3130 LINE(XX,YP)-STEP(XP+XS-XX,YS),C2,BF : [ ノコリヲ ショウキョ
3140 GOTO 2610
3150 '
3160 '
3170 '<<< Selector >>>
3180 MS$ = MS$(0) + ":"
3190 FOR I=1 TO MM
3200 MX(I) = LEN(MS$)*8 + 8 : MW(I) = LEN(MS$(I))*8+4
3210 MS$ = MS$+" "+MS$(I)
3220 NEXT
3230 GOSUB 2250 : MY = WY+1 : MH = WH - 3 : Z = M : [ MヲZニホソク
3240 GOSUB 3280 : IF M < 0 THEN M = Z : Z = -1 : [ ESCナラMヲフック
3250 GOTO 2450 : [ ウイントウヲトシテショウリョウ
3260 '
3270 '
3280 '<<< Move Cursor >>>
3290 GOSUB 3360
3300 K$ = INKEY$ : IF K$ = "" GOTO 3300 ELSE K = ASC(K$)
3310 IF K = 13 THEN RETURN ELSE IF K = 27 THEN M = -1 : RETURN
3320 IF K = 28 THEN K = +1 : GOTO 3340
3330 IF K = 29 THEN K = -1 : GOTO 3340 ELSE 3300
3340 GOSUB 3360 : M = M+K : IF M < 1 THEN M = MM ELSE IF M > MM THEN M = 1
3350 GOSUB 3360 : GOTO 3300
3360 LINE(WX+MX(M),MY)-STEP(MW(M),MH),XO,BF,XOR
3370 RETURN
3380 '
3390 '
3400 '<<< Save >>>
3410 SV = 1 : GOSUB 3670 : GOSUB 2680 : IF F$ = "" THEN RETURN
3420 ER = 0 : C3 = C1 : COLOR,,C3 : BSAVE F$,O,SA,S
3430 C3 = C2 : COLOR,,C3 : IF ER THEN GOSUB 4080 : GOTO 3400
3440 RETURN
3450 '
3460 '
3470 '<<< Load >>>
3480 SV = 0 : GOSUB 3670 : GOSUB 2680 : IF F$ = "" THEN RETURN
3490 ER = 0 : C3 = C1 : COLOR,,C3 : BLOAD F$,S : PUT SPRITE 0,(0,216)
3500 C3 = C2 : COLOR,,C3 : IF ER THEN GOSUB 4080 : GOTO 3470
3510 COPY (XP,YP)-STEP(XS,YS),0 TO (XP,YP),1
3520 RETURN
3530 '
3540 '
3550 '<<< Error Trap >>>
3560 ER = ERR : IF ERL = 1770 THEN RESUME NEXT : [ _SCANノESCチュウタン
3570 IF ER = 53 THEN ER$ = "ファイルカ アリマセン" : RESUME NEXT
3580 IF ER = 56 THEN ER$ = "ファイルメイ エラー" : RESUME NEXT
3590 IF ER = 62 THEN ER$ = "ドライブメイ エラー" : RESUME NEXT
3600 IF ER = 65 THEN RESUME NEXT
3610 IF ER = 66 OR ER = 67 THEN ER$ = "ディスクカ イッハイテス" : RESUME NEXT
3620 IF ER = 68 THEN ER$ = "ライトプロテクトヲハスシテクタサイ" : GOSUB 4080 : RESUME
3630 IF ER = 70 THEN ER$ = "ディスクヲイレテクタサイ" : GOSUB 4080 : RESUME
3640 GOTO 1430
3650 '
3660 '
3670 '<<< Input File Name >>>

```

```

3680 IF TX = 0 THEN SCREEN 0 : COLOR 1,14,14 : WIDTH 40 : TX = 1
3690 FX$ = ".SC"+HEX$(SM)
3700 IF FD$ <> "A:" AND FD$ <> "B:" THEN FD$ = "A:"
3710 PRINT STRING$(11,"-")<< FILES "FD$*" +FX$+" >>STRING$(11,"-");
3720 ER = 0 : FILES FD$+"*." +FX$
3730 IF ER = 0 OR ER = 53 GOTO 3760
3740 BEEP : PRINT"# "ER$ ("FD$")" : FD$ = "A:"
3750 GOTO 3710
3760 IF POS(0) <> 0 THEN PRINT
3770 PRINT STRING$(40,"-");
3780 ER$ = ""
3790 '
3800 '
3810 '<<< Input Again >>>
3820 IF POS(0) <> 0 THEN PRINT
3830 IF ER$ <> "" THEN GOSUB 4080
3840 IF SV THEN PRINT"> SAVE ファイル = "; ELSE PRINT"> LOAD ファイル = ";
3850 F$ = "" : LINE INPUT F$
3860 IF F$ = "" THEN RETURN
3870 FOR I=1 TO LEN(F$) : K$ = MID$(F$,I,1) '[ ファイル メイ ラ オオモシ' ニ ヘンカン
3880 IF "a" =< K$ AND K$ =< "z" THEN MID$(F$,I,1) = CHR$(ASC(K$)-&H20)
3890 NEXT
3900 D$ = FD$
3910 IF MID$(F$,2,1) = ":" THEN D$ = LEFT$(F$,2) : F$ = MID$(F$,3)
3920 IF F$ = "" THEN FD$ = D$ : GOTO 3710 '[ FILES ノ ト' ライフ' ヘンコウ
3930 I = INSTR(F$,".")
3940 IF I > 0 THEN F$ = LEFT$(F$,I-1) '[ カクテョウシ' フトル
3950 F$ = D$+F$+FX$ : ER = 0 : NAME F$ AS F$
3960 IF ER = 56 GOTO 4040
3970 IF SV GOTO 3990 '[ NAME F$ AS F$ ハ カナラス' ERROR
3980 IF ER = 65 THEN RETURN ELSE 4040 '[ LOAD : File Already Exist OK
3990 IF ER = 53 THEN RETURN '[ SAVE : File Not Found OK
4000 IF ER <> 65 GOTO 4040
4010 PRINT"> "F$ ハ ステ' ニ アリマス. OK(y/n) ?"CHR$(29); : GOSUB 1710
4020 K$ = INPUT$(1) : I = INSTR( YN$, K$ ) : IF I = 0 GOTO 4020
4030 IF I < 3 THEN RETURN ELSE PRINT"n" : GOTO 3710
4040 ER$ = ER$+" ("F$+)"
4050 GOTO 3810
4060 '
4070 '
4080 '<<< Hit Any Key >>>
4090 BEEP : TT = TX '[ TT = テキスト モート' フラク' フ ホソソ
4100 IF TX = 0 THEN SCREEN 0 : WIDTH 40 : TX = 1 '[ テキスト モート' ニ スル
4110 IF POS(0) <> 0 THEN PRINT
4120 COLOR 6 : PRINT "# "ER$ : PRINT
4130 PRINT" ((( Hit Any Key )))"; : GOSUB 1710
4140 IF INKEY$ = "" GOTO 4140
4150 PRINT CHR$(13)SPACE$(39)CHR$(13)CHR$(30); '[ Hit Any Key フ ケス
4160 ER = 0 : ER$ = "" : IF TT = 0 GOTO 2680 '[ ク' ラフイック モート' フ フッカツ' サセル
4170 COLOR 1,14 '[ モトモト テキスト モート' タ' ッタ
4180 RETURN
4190 '
4200 '
4210 '<<< Initialize Palette >>>
4220 VDP(9) = VDP(9) OR &H20 : RESTORE 4290
4230 FOR C = 0 TO 15
4240 READ R,G,B
4250 COLOR = ( C, R,G,B )
4260 NEXT
4270 RETURN
4280 '
4290 DATA 7,7,7, 0,0,0, 4,1,1, 6,6,7, 1,1,7, 2,3,7, 5,1,1, 2,6,7
4300 DATA 7,1,1, 7,3,3, 6,6,1, 6,6,4, 1,4,1, 6,2,5, 5,5,5, 7,7,7
4310 '
4320 '
4330 '<<< Initialize Color Table for SCREEN 8 & Set Function Keys >>>
4340 DIM CT(15) : RESTORE 4290

```



〒101 東京都千代田区神田須田町2-6-5 OSビル5F ☎03-252-5581代